[bookmark: R2B1]SysCmd-Funktionen optimal nutzen
Teil 1

SysCmd() fasst mehrere vielseitige und nützliche Funktionen zusammen, die Sie für die unterschiedlichsten Aufgabenstellungen und Problemlösungen einsetzen können. In diesem mehrteiligen Tipp zeigen wir, was dieses Multitalent im Einzelnen zu bieten hat und wie Sie SysCmd() in Ihren Datenbankanwendungen einsetzen können, um den Dialog mit den Anwendern Ihrer Datenbanken zu optimieren, sich die Entwicklung eigener Funktionen zu sparen oder unbeabsichtigten Datenverlusten vorzubeugen.

Beginnen wir mit der Syntax von SysCmd(): Beim Aufruf der Funktion übergeben Sie jeweils eine interne Konstante, die der Funktion mitteilt, was sie machen soll. Um zum Beispiel den Ordner zu bestimmen, in dem Sie Access installiert haben, verwenden Sie die Anweisung

strDir = SysCmd (acSysCmdAccessDir)

Die Konstante "acSysCmdAccessDir" teilt SysCmd() mit, was von ihr erwartet wird, und das heißt in diesem Fall, dass die Funktion das Installationsverzeichnis liefern soll.

Das ist zunächst nur ein einfaches Beispiel; in den meisten Fällen müssen Sie mehr als nur eine einzige Konstante übergeben.

Zunächst zum Thema "Statuszeile": Sie haben zwei Möglichkeiten, um die Statuszeile mit der SysCmd()-Funktion zu beeinflussen: Sie zeigen statischen Text wie "Daten werden importiert..." in der Statuszeile an. Dabei müssen Sie dafür sorgen, dass die Meldung nach Abschluss Ihrer Routine wieder entfernt und die Statuszeile für Access-interne Meldungen freigegeben wird. Die zweite Möglichkeit besteht darin, die Statuszeile mit Hilfe von SysCmd() als Fortschrittsanzeige zu nutzen, um dem Anwender den aktuellen Status eines zeitintensiven Vorgangs anzuzeigen. In diesem Fall muss Ihr Programmcode die Anzeige initialisieren, aktualisieren und auch wieder entfernen.

Für die Ausgabe von statischem Text rufen Sie SysCmd() einmal auf, um den Text anzuzeigen und ein zweites Mal, um den Text wieder zu entfernen. Anfangs übergeben Sie dabei die Konstante "acSysCmdSetStatus" gemeinsam mit dem Text, den Sie anzeigen möchten. Wenn Sie den Text wieder löschen wollen, übergeben Sie der Funktion die Konstante "acSysCmdClearStatus". Das folgende Beispiel zeigt statischen Text an und entfernt ihn wieder, wenn Sie die Meldungsbox bestätigen:

Sub SysCmdTest()

   SysCmd acSysCmdSetStatus, _
          "Text in Statuszeile!"
   MsgBox "Dies ist ein SysCmd()-Test..."
   SysCmd acSysCmdClearStatus

End Sub

Neben der Anzeige von Text in der Statusleiste kann die Funktion SysCmd() während des Ablaufs zeitaufwendiger Vorgänge eine Fortschrittsanzeige in der Statuszeile einblenden. Dazu sind drei Schritte erforderlich:
1. Initialisieren: Um die Funktion für die Fortschrittsanzeige zu initialisieren, übergeben Sie die Konstante "acSysCmdInitMeter", den Text, der links neben der Fortschrittsanzeige erscheinen soll, und den Maximalwert, bei dessen Erreichen die Anzeige 100% entsprechen soll. Alle Aktualisierungen der Fortschrittsanzeige werden dann in Relation zu diesem Maximalwert gesetzt und so der aktuelle Status ermittelt und angezeigt.
1. Aktualisieren: Eine Aktualisierung erfolgt über die Konstante "acSysCmdUpdateMeter". Wenn Sie beispielsweise bei der Initialisierung den Maximalwert 3000 festgelegt haben, muss der Aktualisierungswert zwischen 0 und 3000 liegen. Bei dem Wert 1500 würde SysCmd dann entsprechend einen Fortschritt von 50% anzeigen.
1. Entfernen: Die Fortschrittsanzeige wird entfernt, indem Sie SysCmd() die Konstante "acSysCmdRemoveMeter" übergeben. Ohne diesen Schritt würde Access die Fortschrittsanzeige weiterhin mit "100% erledigt" anzeigen, obwohl die Aufgabe längst abgeschlossen ist.
Alle drei Aufgaben lassen sich mit einer einzigen Prozedur "Hinweis" abdecken, die Sie zunächst in ein beliebiges Modul kopieren:

Sub Hinweis(strMsg As String, Anz As Long)
  Dim X As Variant

  On Error Resume Next
  If Left$(strMsg, 1) = "-" Then 'Zurücksetzen
      X = SysCmd(acSysCmdRemoveMeter)
  ElseIf Left$(strMsg, 1) = "+" Then 'Initialisieren
      X = SysCmd(acSysCmdInitMeter, Mid$(strMsg, 2), Anz)
  Else 'Aktualisieren
      X = SysCmd(acSysCmdUpdateMeter, Anz)
  End If
  DoEvents

End Sub

Am Anfang einer Routine rufen Sie diese Prozedur beispielsweise wie folgt für die Initialisierung auf:

.....
rs.MoveLast
RecCnt = rs.RecordCount
rs..MoveFirst
Hinweis "+Daten werden importiert...", RecCnt
.....

Hier wird ermittelt, wie viele Datensätze ein Recordset umfasst und dieser Wert als Maximalwert mit dem Text "Moment bitte, Daten werden importiert..." an die Prozedur übergeben. Am "+"-Zeichen erkennt die Prozedur, dass zunächst eine Initialisierung der Statuszeile erfolgen soll (acSysCmdInitMeter).

Innerhalb der Schleife erfolgt der Aufruf dann zum Beispiel so:

For I = 1 To RecCnt
  Hinweis "", I
  .....
Next I

Dabei wird nur der aktuelle Schleifenwert gemäß Zählervariable "I" an die Prozedur übergeben. An der leeren Zeichenkette als ersten Parameter erkennt die Prozedur, dass nur der aktuelle Status aktualisiert werden soll (acSysCmdUpdateMeter).

Nach Durchlauf der Schleife wird die Statuszeile wie folgt zurückgesetzt:

Hinweis "-", 0

Hier erkennt die Prozedur am ersten Parameter "-", dass die Statuszeile auf den normalen Status eingestellt werden soll (acSysCmdRemoveMeter).

Weitere nützliche SysCmd()-Funktionen, mit denen Sie beispielsweise Access-Informationen abfragen können, stellen wir in der nächsten Ausgabe vor.

SysCmd() fasst mehrere vielseitige und nützliche Funktionen zusammen, die Sie für die unterschiedlichsten Aufgabenstellungen und Problemlösungen einsetzen können. In diesem mehrteiligen Tipp zeigen wir, was dieses Multitalent im Einzelnen zu bieten hat und wie Sie SysCmd() in Ihren Datenbankanwendungen einsetzen können, um den Dialog mit den Anwendern Ihrer Datenbanken zu optimieren, sich die Entwicklung eigener Funktionen zu sparen oder unbeabsichtigten Datenverlusten vorzubeugen.

Neben den im ersten Teil vorgestellten Funktionen zur Steuerung der Statusleiste kann SysCmd() verschiedene Informationen über die Access-Umgebung liefern. Am effektivsten ist es, wenn Sie anstelle der etwas unhandlichen SysCmd-Aufrufe mit ihren schwierig zu merkenden Konstanten fertige Funktionen verwenden Ein Beispiel für eine solche Hilfsfunktion ist "AccessDir()". Sie ermittelt den Pfad des Ordners, in dem die Programmdatei von Access (MSACCESS.EXE) installiert ist. Dabei endet der Pfadname immer mit einem Backslash "\":

Function AccessDir()

  AccessDir = SysCmd(acSysCmdAccessDir)

End Function

Die Funktion SysCmd() gibt Ihnen auch die Möglichkeit, die verwendete Access-Version zu ermitteln: Der entsprechende Aufruf verwendet die Konstante "acSysCmdAccessVer" und liefert folgende Ergebnisse

"15.0" für Access 2013
"14.0" für Access 2010
"12.0" für Access 2007
"11.0" für Access 2003
"10.0" für Access 2002/XP

In eine Funktion gekleidet, die ein numerisches, leichter abfragbares Ergebnis liefert, sieht das dann so aus:

Function AccessVersion() As Integer

  AccessVersion = Val(Syscmd(acSysCmdAccessVer))

End Function 

Die Funktion konvertiert die Rückmeldung von SysCmd() per "Val()" in einen numerischen Wert, den Sie zum Beispiel per "If AccessVersion() < 10 Then..." abfragen können.

Mit Hilfe von SysCmd() lässt sich auch feststellen, ob die Datenbank mit einer Runtime-Version von Access eingesetzt wird. Dabei hilft Ihnen die Funktion "Runtime()":

Function Runtime()

  Runtime = SysCmd(acSysCmdRuntime)

End Function

Die Funktion liefert als Ergebnis entweder "True/Wahr" oder "False/Falsch". Falls Sie lieber einen Text als Funktionsergebnis zurückliefern möchten, können Sie die folgende Variante der Funktion einsetzen:

Function Runtime()
  Dim varRuntime As Variant

  varRuntime = SysCmd(acSysCmdRuntime)
  If varRuntime = 0 Then
    varRuntime = "Keine Runtime-Version"
  Else
    varRuntime = "Runtime-Version"
  End If
  Runtime = varRuntime

End Function

Wenn Sie ältere Access-Versionen oder Datenbanken im Kompatibilitätsmodus unter neueren Access-Versionen einsetzen und den Arbeitsgruppenmodus nutzen, kann die folgende Funktion hilfreich sein: Sie liefert Namen und Pfad der aktuellen Arbeitsgruppendatei:

Function AccessWorkGroup()

  AccessWorkGroup = SysCmd(acSysCmdGetWorkGroupFile)

End Function

Weitere nützliche SysCmd()-Funktionen, mit denen Sie beispielsweise Objekte überwachen können, stellen wir im dritten Teil in der nächsten Ausgabe vor.

SysCmd() fasst mehrere vielseitige und nützliche Funktionen zusammen, die Sie für die unterschiedlichsten Aufgabenstellungen und Problemlösungen einsetzen können. In diesem mehrteiligen Tipp zeigen wir, was dieses Multitalent im Einzelnen zu bieten hat und wie Sie SysCmd() in Ihren Datenbankanwendungen einsetzen können, um den Dialog mit den Anwendern Ihrer Datenbanken zu optimieren, sich die Entwicklung eigener Funktionen zu sparen oder unbeabsichtigten Datenverlusten vorzubeugen.

Im ersten und zweiten Teil dieses Tipps haben wir Ihnen Grundlagen zur Funktion "SysCmd()" vermittelt und eine Reihe interessanter Funktionen vorgestellt. Ein drittes Einsatzgebiet von SysCmd() betrifft die Abfrage verschiedener Statusinformationen zu beliebigen Datenbankobjekten. So können Sie zum Beispiel feststellen, ob ein Objekt neu, geöffnet, geändert oder ungespeichert ist. Um beispielsweise zu testen, ob ein Objekt in Access geöffnet ist, verwenden Sie die Anweisung

intStatus = SysCmd(acSysCmdGetObjectState, _
                   intObjType, _
                   strObjName)

Wenn Sie beispielsweise den Status eines Moduls mit dem Namen "modInfo" abfragen möchten, müsste der Funktionsaufruf folgendermaßen lauten:

intStatus = SysCmd(acSysCmdGetObjectState, _
                   acModule, _
                   "modInfo")

Falls es sich bei der Rückmeldung um den Wert "0" handelt, ist das Objekt nicht geöffnet. Wenn ein anderer Wert als Null vorliegt, kann der zurückgelieferte Wert die Summe beliebiger oder sogar aller Konstanten sein:

Geöffnet: acObjStateOpen
Neu: acObjStateNew
Geändert, aber nicht gespeichert: acObjStateDirty

Wenn ein Objekt neu ist, besteht der zurückgelieferte Wert beispielsweise aus der Summe von "acObjStateOpen" und "acObjStateNew". Sie müssen daher den Operator "AND" verwenden, um die Elemente zu ermitteln, aus denen sich das Ergebnis zusammensetzt. Um festzustellen, ob das Objekt "frmTest" neu ist, verwenden Sie die folgenden Anweisungen:

Dim intStatus As Integer
...
intStatus = SysCmd(acSysCmdGetObjectState, _
                   acForm, _
                   "frmTest")
If intStatus AND acObjStateNew Then
  'Neues Objekt
End If
...

Sie können den Status nicht mit "acObjStateNew" vergleichen, da der Status in Wirklichkeit die Summe aus "acObjStateOpen", "acObjStateNew" und "acObjStateDirty" ist. Wenn Sie den von der Funktion SysCmd() zurückgelieferten Wert mit der Konstanten "acSysCmdGetObjectState" kontrollieren, liefert Access den Wert Null, falls das Objekt nicht geöffnet ist, und in allen anderen Fällen einen Wert größer als Null.

Mit Hilfe des folgenden Codes können Sie feststellen, ob ein Formular geöffnet ist. Die Funktion "IsFormOpen()" liefert entweder den Wert "True/Wahr" oder "False/Falsch":

Function IsFormOpen(strName As String) As Boolean

 IsFormOpen = (SysCmd(acSysCmdGetObjectState, _
                      acForm, strName) <> 0)
End Function

Wenn Sie Informationen über ein Objekt abfragen möchten, bei dem es sich nicht um ein Formular handelt, übergeben Sie den Objekttyp (acForm, acTable, acQuery etc.) als optionalen Parameter und können so die Abfrage in einer einzigen Funktion zusammenfassen:

Function IsOpen(strName As String, _
                Optional varObjType As Variant) _
                As Boolean

   If IsMissing(varObjType) Then
      varObjType = acForm
   End If

   IsOpen = (SysCmd(acSysCmdGetObjectState, _
                    varObjType, strName) <> 0)
End Function

Falls beim Aufruf der Funktion kein Objekttyp festgelegt wird, geht die Funktion vom Objekt "acForm" aus.

In vielen Datenbanken muss eine VBA-Routine solange unterbrochen werden oder eine Pause machen, bis ein zuvor geöffnetes Formular oder ein in der Seitenansicht angezeigter Bericht wieder geschlossen ist. Mit der Anweisung

DoCmd.OpenForm "Neuer Datensatz", , , , ,acDialog

lässt sich hier nur bedingt arbeiten, da sie nicht mit Berichten angewendet werden kann und nicht immer alle Formulare als Dialoge angezeigt werden können.

Sie können hier zur Lösung des Problems auf "GetObjectState()" von "SysCmd()" zurückgreifen: Über diese Sub-Funktion von SysCmd() lässt sich einfach feststellen, ob ein Formular oder ein Bericht geöffnet ist und solange in einer entsprechenden Schleife warten.

Sub WaitForm(strForm As String)

  While SysCmd(acSysCmdGetObjectState, _
               acForm, strForm)
    DoEvents
  Wend

End Sub

Sub WaitReport(strReport As String)

  While SysCmd(acSysCmdGetObjectState, _
               acReport, strReport)
    DoEvents
  Wend

End Sub

Die Funktion SysCmd() prüft hier jeweils über die Sub-Funktion "GetObjectState", ob das Objekt, dessen Name als Parameter übergeben wurde, geladen ist. Die Anweisung "DoEvents" sorgt dafür, dass Access innerhalb der Schleife Prozessorzeit an das System zurückgibt und so die Performance nicht bremst.

Diese beiden Prozeduren können Sie in VBA-Routinen einsetzen, um mit der Ausführung zu warten, bis ein Formular oder Bericht wieder geschlossen ist:

Private Sub btnTestForm_Click()

  DoCmd.OpenForm "Kunden"
  WaitForm "Kunden"
  MsgBox "Wieder da!"

End Sub

Hier wird zunächst ein Formular geöffnet und anschließend die Prozedur "WaitForm" mit dem Namen des betreffenden Formulars als Parameter aufgerufen. "WaitForm" wartet nun solange, bis das Formular geschlossen ist, anschließend wird die VBA-Routine weiter ausgeführt. Sie zeigt hier zu Testzwecken eine kleine Meldung an. Genauso einfach warten Sie auch darauf, dass ein Bericht wieder geschlossen wird:

Private Sub btnTestReport_Click()

  DoCmd.OpenReport "Katalog", acPreview
  WaitReport "Katalog"
  MsgBox "Wieder da!"

End Sub

